Set 5, Part 1: Symbolic Algebra with *Mathematica*

To explore the functionality of *Mathematica,* I'll use it to solve a problem from my Ph1c set last week.

Consider a circuit with a battery of voltage *V*, resistor with resistance *R,* and inductor with inductance *L* in series. The voltage around the single loop must sum to 0, which gives us an ODE for the current *j:*

```
In[72]:= ODE = {V - j[t] R - j'[t] L == 0}
```

Out[72]= $\{-L\, j'(t) - R\, j(t) + V = 0\}$

```
In[73]:= DSolve[ODE, {j[t]}, {t}]
```

Out[73]= $\left\{\left\{j(t) \to c_1\, e^{-\frac{R t}{L}} + \frac{V}{R}\right\}\right\}$

We define a function for the current *J(t)* according to this solution:

In[74]:= $\texttt{J[t\_] :=}\ c_1\, e^{-\frac{R t}{L}} + \frac{V}{R}$

We know that at *t = 0*, the inductor will fight the current change completely. Thus, *J(0) = 0.*

In[75]:= $\texttt{Solve[J[0] == 0, \{}c_1\texttt{\}]}$

Out[75]= $\left\{\left\{c_1 \to -\frac{V}{R}\right\}\right\}$

Now that we have the complete solution, let us store it as the actual current *i(t):*

In[76]:= $\texttt{i[t\_] = J[t] /.}\ c_1 \to -\frac{V}{R}$

Out[76]= $\frac{V}{R} - \frac{V\, e^{-\frac{R t}{L}}}{R}$

Now suppose we allow the circuit to run until some time *T.* To calculate the energy delivered by the battery over this time, we integrate power:

In[77]:= $\texttt{batE =}\ \int_0^T \texttt{i[t] V}\, dt$

Out[77]= $\frac{V^2\left(L\left(e^{-\frac{R T}{L}} - 1\right) + R\, T\right)}{R^2}$

The energy stored in the inductor at time *T* is simply:

In[78]:= $\texttt{inductorE =}\ \frac{1}{2}\texttt{L (i[T])}^2$

Out[78]= $\frac{1}{2} L \left(\frac{V}{R} - \frac{V\, e^{-\frac{R T}{L}}}{R}\right)^2$

And the energy dissipated by the resistor over time is also obtained from integrating power. However, the resistor's voltage changes over time:

In[79]:=
```
resistorE = ∫_0^T (i[t])^2 R ⅆt
```

Out[79]=
$$\frac{V^2\left(L\,e^{-\frac{2RT}{L}}\left(4\,e^{\frac{RT}{L}}-1\right)-3L+2RT\right)}{2R^2}$$

We confirm that energy was conserved:

In[80]:=
```
Ediff = batE - (inductorE + resistorE)
```

Out[80]=
$$\frac{V^2\left(L\left(e^{-\frac{RT}{L}}-1\right)+RT\right)}{R^2}-\frac{V^2\left(L\,e^{-\frac{2RT}{L}}\left(4\,e^{\frac{RT}{L}}-1\right)-3L+2RT\right)}{2R^2}-\frac{1}{2}L\left(\frac{V}{R}-\frac{V\,e^{-\frac{RT}{L}}}{R}\right)^2$$

In[81]:=
```
Simplify[Ediff]
```

Out[81]= 0